37/replication_package/Do/Figures/Fig10.do

**********
* Immigration and Redistribution
* Alesina, Miano, Stantcheva
* RESTUD

* Figure 10
**********

clear all

* Specify directory of the replication package
global dir "/Users/armandomiano/Dropbox/AMS_Redistribution/Data/Survey_Data/Replication RESTUD sharing"
cd "$dir"

* Declare output sub-directory

global outdir "Out/Figures"

* Load the data

use "Data/survey_analysis.dta", clear

* Countries included in the analysis
global countries "US UK DE FR IT SE"


* Generate flags - To ensure answers quality
gen flag_1=0
gen flag_2=0

bysort country treatment_recod: egen min_duration = pctile(duration), p(2)
bysort country treatment_recod: egen max_duration = pctile(duration), p(98)

replace flag_1=1 if duration<min_duration
replace flag_2=1 if duration>max_duration

* Drop respondents that have spent too little or too much time (bottom/top 2%)

keep if flag_1==0 & flag_2==0

*Drop variable used to exclude inattentive respondents
drop flag_1 flag_2 min_duration max_duration

* Add data on actual natives and immigrants statistics and generate misperceptions *

do "Do/misperceptions.do"

* Gen additional variables entering the indices *

* Dummy = 1 if Mohammad receives more on net from the state (either receives more transfers and pays same or less taxes or receives same transfers but pays less taxes)
gen moh_more=(Moh_transfers<3 & Moh_tax>2)
replace moh_more=1 if Moh_transfers==3 & Moh_tax>3
replace moh_more=. if Moh_tax==. | Moh_transfers==.
la var moh_more "Mohammad receives more transfers and/or pay less taxes than John"
label val moh_more binary

* Transfers
gen imm_tra_more=(transfers_imm==10 |transfers_imm==11 |transfers_imm==12 |transfers_imm==13)
la var imm_tra_more "Immigrants receive more government transfers than natives"
replace imm_tra_more=. if transfers_imm==.
label val imm_tra_more binary

* Immigration is not a problem
gen imm_not_problem=(q_imm_problem==16 | q_imm_problem==17)
replace imm_not_problem=. if q_imm_problem==.
la var imm_not_problem "Immigration is not a problem"
label val imm_not_problem binary

* When should immigrants be entitled to get benefits?
gen imm_benefits_soon=(q_imm_benefits==1 | q_imm_benefits==2 | q_imm_benefits==4)
replace imm_benefits_soon=. if q_imm_benefits==.
la var imm_benefits_soon "Immigrants should get benefits in less than 3 years"
label val imm_benefits_soon binary

* When should immigrants be allowed to apply for citizenship?
gen imm_citizenship_soon=(q_imm_citizenship==1 | q_imm_citizenship==2)
replace imm_citizenship_soon=. if q_imm_citizenship==.
la var imm_citizenship_soon "Immigrants should get citizenship in 2 or 5 years"
label val imm_citizenship_soon binary

* When would you consider an immigrant "Truly American"?
gen trully_american_cit=(q_imm_american==7 | q_imm_american==1 | q_imm_american==4)
replace trully_american_cit=. if q_imm_american==.
la var trully_american_cit "Consider American at citizenship or sooner"
label val trully_american_cit binary

*Budget allocation
winsor2 budget_safetynet budget_health budget_education, s(_w) c(5 95) by(country)
la var budget_safetynet_w "% of the budget assigned to Income Support Program (winsorized)"
la var budget_health_w "% of the budget assigned to Public Spending on Health (winsorized)"
la var budget_education_w "% of the budget assigned to Spending on Schooling (winsorized)"

gen budget_social_w = budget_health_w + budget_safetynet_w
la var budget_social_w "% of the budget assigned to Income Support Program and Health (winsorized)"

* Inequality is not a problem
gen ineq_no_problem=(q_inequality_problem==1)
replace ineq_no_problem=. if q_inequality_problem==.
la var ineq_no_problem "Inequality is not a problem"
label val ineq_no_problem binary

* Total donation above 50th percentile within country
gen total_donation_d=0
foreach x in $countries {
su total_donation if country=="`x'", d
replace total_donation_d=1 if total_donation>r(p50) & country=="`x'"
}
replace total_donation_d=. if total_donation==.

la var total_donation_d "Donated more than the median of the country of reference"
label val total_donation_d binary


*** Keep only control group ***

keep if control==1


****************
* Gen controls *
****************


* Gen Left-right variables *
* Based on vote or voting intensions
gen left=(party_voted==4 | party_voted==5)
replace left=. if party_voted==. | party_voted==6 | party_voted==0
gen right=(party_voted==1 | party_voted==2)
replace right=. if party_voted==. | party_voted==6 | party_voted==0
gen center=(party_voted==3)
replace center=. if party_voted==. | party_voted==6 | party_voted==0

label var left "Left-wing"
label var right "Right-wing"
label var center "Center"

foreach i in left right center{
label val `i' binary
}


* Young
gen young=(age<45)
label var young "Young"
label val young binary

* Gender
gen male=(sex==1)
label var male "Male"
label val male binary

* Children dummy
gen children=(number_children>1)
replace children=. if number_children==.
label var children "Has children"
label val children binary

* Immigrant parent
gen immigrant_parent=(q_parent_same==2)
replace immigrant_parent=. if q_parent_same==.
label var immigrant_parent "At least one of the parents is an immigrant"
label val immigrant_parent binary

* Top income
gen top_income=0
foreach x in $countries{
su household_income if country=="`x'", d
replace top_income=1 if household_income>r(p75) & country=="`x'"
}
label var top_income "High Income"
label val top_income binary



* Redistribution index

foreach var in tax_top1 tax_bottom50 budget_social_w budget_education_w ineq_no_problem total_donation_d{
				
su `var'
local `var'mc: display %5.3f `r(mean)'
su `var'
local `var'sdc: display %5.3f `r(sd)'
gen `var'_ind=(`var'-``var'mc')/``var'sdc'
replace `var'_ind=(``var'mc'-``var'mc')/``var'sdc' if `var'==.
}

gen red_index = (tax_top1_ind - tax_bottom50_ind + budget_social_w_ind + budget_education_w_ind - ineq_no_problem_ind + total_donation_d_ind)/6

foreach var in tax_top1 tax_bottom50 budget_social_w budget_education_w ineq_no_problem total_donation_d{
	drop `var'_ind
}

*** Gen Immigration support index

foreach var in imm_not_problem imm_benefits_soon imm_citizenship_soon ///
						trully_american_cit q_govt_imm  {
su `var'
local `var'mc: display %5.3f `r(mean)'
su `var'
local `var'sdc: display %5.3f `r(sd)'
gen `var'_ind=(`var'-``var'mc')/``var'sdc'
replace `var'_ind=(``var'mc'-``var'mc')/``var'sdc' if `var'==.
}

gen index_imm_support=(imm_benefits_soon_ind + imm_citizenship_soon_ind ///
						+trully_american_cit_ind + q_govt_imm_ind +imm_not_problem_ind)/5					

				
drop imm_not_problem_ind imm_benefits_soon_ind imm_citizenship_soon_ind ///
						trully_american_cit_ind q_govt_imm_ind

******************************
* Perception indices    *****
******************************
*** Gen indices ***

foreach var in mis_share_mu mis_share_ch mis_share_LA mis_share_AS  mis_share_AF mis_share_E  mis_share_NA ///
				mis_unemp_imm  mis_loweduc_imm  mis_higheduc_imm mis_poverty_imm ///
				effort_poor  moh_more  imm_tra_more{
				
su `var'
local `var'mc: display %5.3f `r(mean)'
su `var'
local `var'sdc: display %5.3f `r(sd)'
gen `var'_ind=(`var'-``var'mc')/``var'sdc'
replace `var'_ind=(``var'mc'-``var'mc')/``var'sdc' if `var'==.
}

gen culture_index = (mis_share_mu_ind - mis_share_ch_ind + mis_share_LA_ind + mis_share_AS_ind + mis_share_AF_ind - mis_share_E_ind - mis_share_NA_ind)/7

gen econ_index = (mis_unemp_imm_ind + mis_loweduc_imm_ind - mis_higheduc_imm_ind +mis_poverty_imm_ind)/4

gen free_riding_index = (effort_poor_ind + moh_more_ind + imm_tra_more_ind)/3

foreach var in mis_share_mu mis_share_ch mis_share_LA mis_share_AS  mis_share_AF mis_share_E  mis_share_NA ///
				mis_unemp_imm  mis_loweduc_imm  mis_higheduc_imm mis_poverty_imm ///
				effort_poor  moh_more  imm_tra_more{
					
				drop `var'_ind
				
				}

				
label var culture_index "Perc. cultural distance index"
label var econ_index "Perc. economic weakness index"
label var free_riding_index "Perc. free-riding index"
				

********************************************
* Declare variables to be used as controls *
********************************************

global controls right left male young immigrant_parent children university_degree top_income sector_dummy


**************************
* Correlation Graph with support for immigration and redistribution
*************************

global X mis_share_foreign culture_index econ_index free_riding_index

preserve

** Standardize vars **

foreach var in red_index index_imm_support $X $controls{
	su `var'
	local `var'mc: display %5.3f `r(mean)'
	local `var'sdc: display %5.3f `r(sd)'
	replace `var'=(`var'-``var'mc')/``var'sdc'
	}

* Regressions (All together) - Immigration
	eststo: xi: reg index_imm_support $X $controls i.country, robust
foreach var in $X{
	local b`var'_imm= _b[`var']
	local se`var'_imm = _se[`var']
	local ub_b`var'_imm= _b[`var'] + 1.96*_se[`var']
	local lb_b`var'_imm= _b[`var'] - 1.96*_se[`var']     
}

* Regressions (All together) - Redistribution


	eststo: xi: reg red_index $X $controls i.country, robust
foreach var in $X{
	local b`var'_red= _b[`var']
	local se`var'_red = _se[`var']
	local ub_b`var'_red= _b[`var'] + 1.96*_se[`var']
	local lb_b`var'_red= _b[`var'] - 1.96*_se[`var']     
}

* Make Graph

        clear
        set obs 17
        egen t = seq()
		label define quintile 15 "Perc. free-riding index" 11 "Perc. economic weakness index" 7 "Perc. cultural distance index" 3 "All Immigrants (misp.)" 
		label values t quintile
				
        gen bY = .
        gen ub_bY = .
        gen lb_bY = .
		gen bN = .
        gen ub_bN = .
        gen lb_bN = .		
		gen n=_n
		
	
		local q = 3
	    foreach var in $X {	
		replace bY=`b`var'_imm' if n==`q'+1
		replace ub_bY=`ub_b`var'_imm' if n==`q'+1
		replace lb_bY=`lb_b`var'_imm' if n==`q'+1
		local q = `q' + 4
		}
		
		local q = 3
		foreach var in $X {
		replace bN=`b`var'_red' if n==`q'-1
		replace ub_bN=`ub_b`var'_red' if n==`q'-1
		replace lb_bN=`lb_b`var'_red' if n==`q'-1
		local q = `q' + 4
		}

twoway 	(rspike lb_bY ub_bY t, hor lcolor(blue*.2) lpattern(solid) lwidth(vthick)) ///
			(scatter t bY, mcolor(blue*0.7) lcolor(blue*0.7) lpattern(solid) msymbol(S) msize(large)) ///
			(rspike lb_bN ub_bN t, hor lcolor(red*.2) lpattern(solid) lwidth(vthick)) ///
			(scatter t bN, mcolor(red*1.2) lcolor(red*1.2) lpattern(solid) msymbol(D) msize(large)), ///
            ytitle("") xlabel(-0.4[0.1]0.15, labsize(medlarge)) ylabel(3 7 11 15, value labsize(medlarge) angle(0) glcolor(gs16) noticks)  ///
            xtitle("Correlation", size(medlarge)) ///
			legend(order(2 4) row(2) lab(2 "Support for Immigration") lab(4 "Support for Redistribution")) ///
			yline(5, lcolor(gs8) lpattern("-")) yline(9, lcolor(gs8) lpattern("-")) ///
			yline(13, lcolor(gs8) lpattern("-")) ///
			xline(0, lcolor(black) lw(vthin))  ///
            graphregion(color(white)) plotregion(color(white))
			graph export "$outdir/Figure_10.eps", as(eps) replace		

restore